
ABSTRACT 

Standard clinical datasets o�en contain character variables with texts. Some of the examples of such 
variables include adverse events terms, disposi�on reason and comment fields. There is o�en a 
requirement to have prespecified capitaliza�on on these variables. Moreover, the raw datasets, which 
are the primary source for these datasets o�en contains text strings with typographical errors and 
random capitaliza�on.  Use of standard SAS func�ons such as propcase or lowcase do not always give the 
desired capitaliza�on. Similarly, use of characters replacement func�ons such as tranwrd and translate 
func�ons to fix capitaliza�on and typographical errors can be cumbersome when mul�ple words are 
involved. To address this shortcoming, a single macro program called “Fontcase Macro” has been created 
which performs the following tasks simultaneously with minimal user input: (i) converts texts of any 
prior capitaliza�on to sentence case or �tle case as required, (ii) converts or maintains 
abbrevia�ons/standard terminologies to standard capitaliza�on, (iii) corrects typos. The macro code uses 
the Perl Regular Expression (RegEX) func�on prxchange to iden�fy text pa�erns and change them to 
desired capitaliza�on. Moreover, prespecified abbrevia�ons and standard terminologies are stored in a 
spreadsheet file that is accessed by the macro program. The macro has been extremely useful in the 
genera�on of standard clinical datasets such as SDTM (Study Data Tabula�on Model) and ADaM 
(Analysis Data Model) datasets for variables involving text strings and in the genera�on of TLFs (tables, 
lis�ngs, and figures) using raw datasets. 

  
INTRODUCTION 

SAS programmers very frequently come across instances where we are dealing with character variables 
containing lengthy texts and strings.  If these variables are used in the genera�on of output tables or 
standard datasets, it is o�en required to alter the capitaliza�on of these text strings. In clinical trials 
study, standard datasets such as SDTM and ADaM datasets are generated using raw data which contain 
inconsistent capitaliza�on and typos. And trying to find a correct capitaliza�on, or in other instances, 
dealing with typos is a frequent occurrence and is o�en �me consuming. A program that addresses for 
these capitaliza�on requirements and addresses the inaccuracy issues with minimal user input saves 
significant programming �me. The paper presents a macro program, fontcase macro, which was created 
to overcome these issues. The macro incorporates the following tasks: 

1. Convert texts of any capitaliza�on to either sentence case or �tle case as desired. 
2. Adjust standard terms and abbrevia�ons to prespecified capitaliza�on. 
3. Correct typos.  

 
Hence, the macro can be for the sole purpose of correc�ng typos and/or abbrevia�ons, the sole purpose 
changing capitaliza�on, or a combina�on of both.  It also offers the flexibility to take variable values of 
any capitaliza�on and change them to sentence case or �tle case.  
 
The macro primarily uses the SAS func�ons propcase and lowcase in conjunc�on with the powerful Perl 
Regular Expression (Regex) to get the desired capitaliza�on and correct typos and 



abbrevia�ons/standard terms. While the pa�ern matching and replacement ability of Regex helps in the 
replacement of incorrect terms, the pa�ern matching and replacement ability together with posi�ve and 
nega�ve lookbehind and lookahead features of Regex help in assigning correct capitaliza�on by 
iden�fying standalone words and the beginning and end of sentences. In addi�on to the program code, 
the macro reads from a separate spreadsheet containing standard and abbreviated terms with fixed 
capitaliza�on and uses the prxchange Regex func�on to find and replace the texts in character variables. 
The use of a separate spreadsheet ensures minimal user interven�on in the macro program code and 
provides flexibility to the user to make required addi�ons/edits to the list of terms in the spreadsheet on 
a need basis.  
 
MACRO PARAMETERS AND DEFINITION 

The following is the macro with parameters used in fontcase macro: 
 
%fontcase (data =, fcase =, varlist =, oldval =, newval =, listpath =); 
 

 where data is the SAS dataset of interest, 
fcase is the desired capitaliza�on. fcase is assigned a value of propcase for �tle case, lowcase for 
sentence case, and le� blank for no change in exis�ng capitaliza�on, 
varlist is the variable in the dataset you want to run the macro on, 
oldval is the word/string that you want to search and replace. oldval can be mul�ple words separated by 
comma, 
newval is the corresponding replacement value(s) for oldval, 
Listpath is the filepath and filename of the spreadsheet containing terms with prespecified capitaliza�on.  
 
Among the macro variables, fcase, oldval, newval and listpath can assume blank values depending on 
exis�ng variable values. A blank value for fcase denotes that no change in capitaliza�on is required. 
Blank values for oldval and newval denote that there are no typos in the variable filed that needs 
correc�on, and a blank value for listpath denote that no correc�on is required for standard and 
abbreviated terminologies. 
 
Figure 1 shows the flow diagram for the execu�on of fontcase macro. The macro is run with the ini�al 
set of assignments for the macro parameters. If the output variable requires further correc�on a�er the 
ini�al run, then we update the macro parameter accordingly. In most cases, the updates involve adding 
term(s) in the spreadsheet and upda�ng the terms specified in the oldval and newval macro parameters. 
 
 



 
Fig 1. Flow diagram of fontcase macro execution 

Figure 2 shows the snippet of the spreadsheet that contains the standard abbrevia�on terms. The terms 
in the spreadsheet is a custom list for the variable values used in this paper. The spreadsheet is unique to 
our dataset and can be updated any�me during the program run to accommodate for any missing terms. 

 
Fig 2. Snippet of excel spreadsheet containing abbreviated and standard terminologies. 
 
PROGRAMMING CODE AND STEPS 

The following is the sequen�al order of programming steps and code: 

 
Step1: Locate input dataset and data library, if reading from external loca�on. 
 
libname input "C:\temp"; 

Step2: Iden�fy lowercase terms and assign as a macro variable. 
 
***************************************************************************** 



Assign macro variable “lcase” to the terms that are always lowercase even for 
Title case separated by “|”. Macro variable list contains the following parts 
of speech. 

- articles  
 - Short prepositions (3 letter or less) 
 - Short conjunctions (3 letter or less).  
****************************************************************************; 
 
%let lcase = 
%str(a|an|the|as|at|but|by|for|in|of|off|on|per|to|up|via|and|but|for|nor|or|
so|yet);  
 

Step 3: Import standard and abbrevia�on terms from spreadsheet and assign as macro variables. 

***************************************************************************** 

Import standard and abbreviation terms from the Excel spreadsheet. Path and 
filename are specified in the macro parameter “listpath”. The spreadsheet 
contains a single column and lists individual terms in desired capitalization 
in individual rows. The program code looks for the terms on the list 
irrespective of case sensitivity but replaces with the capitalization as 
specified in the spreadsheet. 
****************************************************************************; 
 
proc import datafile="&listpath."  out=temp1 dbms=excel replace; 
 getnames=no; 
run; 
 
 
************************************************************************** 
Remove "/" from the spreadsheet terms, if any since it generates syntax error 
while using the PRXCHANGE statement. The "/" term will be reinstated at a 
later step. 
****************************************************************************; 
 
data temp2; 

set temp1; 
 term=tranwrd(F1, '/', '###'); 
run; 
 
************************************************************************** 
List all the spreadsheet terms as a macro variable “abbrev” separated by “|”. 
****************************************************************************; 
 
proc sql noprint;  

select term into: abbrev separated by "|" from temp2; 
quit; 
%put &abbrev.; 
 
Step 4: Read dataset from input library and perform manipula�ons for desired capitaliza�on. 

data &data.; 
 set input.&data.;  
 options noquotelenmax; 
 %let count1 = %sysfunc(countw(%quote(&abbrev), %str(|) )) ; 
 %let count2 = %sysfunc(countw(%quote(&oldval), %str(,) )) ;    



 %let var1=&varlist.;      
   
***************************************************************************** 
Change all the variable texts to “propcase” or “lowcase”, or leave as is 
depending on the output requirement. 
****************************************************************************; 
 

%if &fcase. = lowcase or &fcase.=propcase %then %do;   
 &var1. = &fcase.(&var1.); 
%end; 

 
***************************************************************************** 
 Change abbreviations/standard terms to prespecified capitalization as listed 
on the spreadsheet. Reinstate the “/” term if it was present and was earlier 
converted to “###” 
****************************************************************************; 
 

&var1.= tranwrd(&var1., '/', '###'); 
%do m = 1 %to &count1.; 

 %let old4 = %scan(&abbrev, &m, "|");  
 &var1. = prxchange("s/\b&old4.\b/&old4./i", -1, &var1.); 
%end; 
   
&var1. = tranwrd(&var1., '###','/' ); 

 
***************************************************************************** 
Change articles, short prepositions, and short conjunctions stored in the 
macro variable “lcase” to lowercase for title case output. Code uses regex 
negative lookbehind and negative lookahead to ensure that only standalone 
terms are replaced. For example, it identifies and changes “or” but not 
“Organ” during title case conversion. 
****************************************************************************; 
 

%if &fcase. = propcase or &fcase.=lowcase %then %do; 
&var1. = prxchange("s/(?<!\w|\/|\')(&lcase.)(?!\w|\/)/\L$1\E/i", 
-1, &var1.);  

%end;  
   
***************************************************************************** 
Capitalize beginning word of a sentence to uppercase for title case and 
sentence case output. Code used uses regex positive lookbehind. 
****************************************************************************; 
 

%if &fcase.= lowcase or &fcase.=propcase %then %do; 
&var1. = 
prxchange('s/((^\s*\w)|(?<=\.|\?|\!|\:)(\s*\w))/\U$1\E/',-
1,&var1); 

%end; 
 
***************************************************************************** 
Capitalize end word of a sentence to uppercase for Title case output. 
***************************************************************************; 
 
 %if &fcase.= propcase %then %do; 
  &var1. = prxchange("s/(\w+)(?=\.|\?|\!|\:)/\u\L$1\E/",-1,&var1); 
 %end; 
    



   
***************************************************************************** 
Change terms containing numbers and characters, such as Subject IDs, dates, 
temperature values etc., to uppercase. Eg, subject ID of 01XYZ2345 or a date 
of 01JAN2024. 
****************************************************************************;  
 

&var1. = prxchange("s/((\d+\-\w+\d?)|(-?\d+\w?)|(\d+\-
\w+)|(\w+\d+))/\U$1\E/i",-1,&var1.); 

 
***************************************************************************** 
Correct Typos by replacing terms assigned to the macro variable “oldval” to 
terms assigned to the macro variable “newval” 
****************************************************************************; 
 
 %do n=1 %to &count2.; 
  %let old5 = %scan(&oldval., &n, ',');  
   %let new5 = %scan(&newval., &n, ',');   
    &var1. = prxchange("s/\b&old5.\b/&new5./i", -1, &var1.);  
 %end;    
 
run; 
 
%mend fontcase; 

Step 5: Macro Call. 
 
***************************************************************************** 
(a) Example of macro call to change CHARVAR variable values to sentence case. 
In addition to changing capitalization, the terms "Bili" and "creat" are 
corrected to "bilirubin" and "creatinine" by assigning parameter values in 
oldval and newval.  
****************************************************************************; 
 
%fontcase (data=test,fcase=lowcase, varlist=%str(CHARVAR), 

oldval=%str(bili, creat), newval=%str(bilirubin, creatinine), 
listpath=%str(&filedir.\abbrevlist.xlsx)); 

 
***************************************************************************** 
(b) Example of macro call to change CHARVAR variable values to title case. 
The terms "Bili" and "creat" are corrected to "Bilirubin" and "Creatinine" by 
assigning parameter values in oldval and newval.  
****************************************************************************; 
 
fontcase (data=test,fcase=propcase, varlist=%str(CHARVAR), 

oldval=%str(bili, creat), newval=%str(Bilirubin, Creatinine), 
 listpath=%str(&filedir.\abbrevlist.xlsx)); 

 
***************************************************************************** 
(c) Example of macro call where capitalization of CHARVAR variable values is 
left unchanged. The only correction here is the capitalization on standard 
abbreviation terms as listed in the "abbrevlist" spreadsheet. 
****************************************************************************; 
 
%fontcase (data=test,fcase=, varlist=%str(CHARVAR), 

oldval=%str(), newval=%str(), 
listpath=%str(&filedir.\abbrevlist.xlsx)); 



OUTPUT 

Figure 3 shows outputs of the conversion from a test dataset containing character variable CHARVAR. 
Sec�on a is the variable in its original form, As seen in the figure, the variable texts are inconsistent in 
terms of capitaliza�on. Among the highlighted terms, “hgb’ and “wbc” should be uppercase since these 
are standard abbreviated medical terms. Similarly, we want to replace the terms “create” and “Bili” with 
“Crea�nine” and “Bilirubin”. Referring to the macro call in Step 5, the abbrevia�ons are corrected using 
the “abbrevlist.xlsx” spreadsheet (Fig 2), and the replacement terms are updated using the oldval and 
newval macro parameters. Sec�on b in figure 3 shows the conversion to sentence case and sec�on c 
shows the conversion to �tle case. In addi�on to changing the texts to desired capitaliza�on, the terms 
men�oned earlier are corrected as highlighted in the figure sec�ons. In the output shown in sec�on d, 
the capitaliza�on is le� unchanged, and the only updates are to the abbreviated terms that are 
capitalized. The output in the figure shows versa�lity and flexibility offered by the macro.  

 
Fig 3. Output of conversion of character variable values to different capitalization. (a) is the variable values in 
original form , (b) is the conversion to sentence case, (c) is the conversion to title case, (d) is the correction of just 
the abbreviation terms. 

 

LIMITATIONS, PITFALL AND REMEDY 

Conversion to sentence case is not recommended if the character strings include lot of proper nouns. 
The macro does not correct for proper nouns, unless specified in the spreadsheet or “oldval”/”newval” 
macro parameters. And the process can be cumbersome if there are a lot of proper names involved.  

As for the pi�alls of using this macro, there are chances of few errors being introduced in the variable 
value because of automa�c conversion of all terms to propcase or lowcase. Such errors can be remedied 
using the following steps: 



- Enter the erroneous term and correc�on term in the “oldval” and “newval” parameters and 
rerun the macro, leaving all the other parameters unchanged in macro invoca�on. 

- Enter correct capitaliza�on for the incorrect terms in the spreadsheet. 
 

CONCLUSION 

The paper provided the detailed code of the fontcase macro and provided examples of usage for 
different condi�ons. As seen from the examples, the macro offers versa�lity to update character 
variables depending on the users’ needs. Although an example was not presented in this paper, the 
macro can also be used equally effec�vely to change upper case texts to sentence/�tle case. Working 
with character variables comes with many challenges in the form of capitaliza�on, typographical and 
gramma�cal issues and may o�en be �me consuming. fontcase macro can be a very useful addi�on to 
tackle such issues.   

